
2. INTRODUCTION

This contribution proposes that the iterators specified in TS 32.111-3 [1] and TS 32.106-6 [1] be changed so that they are consistent with each other. It will be confusing to have different methodologies for handling iterators in the two different CORBA specifications. The iterators are performing the same basic functionality, returning a subset of the results until completion.

The intent is to synchronise the CORBA iterators between the 3GPP SA5 and the current T1M1.5 / ITU-T CORBA frameworks [7].

This document replaces technical document S5C000092 [5].

The different iterators are shown in the next two sections. (Note that editing has been done to remove IDL not directly related to the iterators.)

2.1.  Existing TS 32.111-3 Iterator

/**

The AlarmInformationIterator is used to iterate through a set of Alarm Informations in Alarm List.  Method get_alarm_list contains it as output parameter.

IRPManager uses it to pace the return of Alarm Informations. IRPManager cannot use it to pace when IRPAgent should retrieve Alarm Informations from Alarm List.

*/

interface AlarmInformationIterator {

/**

This method returns up to "how_many" Alarm Informations. If 1 or more Alarm Information is returned, return TRUE. Return FALSE if there is no more Alarm Information to be returned.

*/

boolean next_alarmInformations (

in unsigned long how_many,

out AlarmIRPConstDefs::AlarmInformationSeq alarm_informations

)

raises (NextAlarmInformations,InvalidParameter);

/**

This method destroys the iterator.

*/

void destroy ();

}; // end of AlarmInformationIterator

/*

This interface specifies all methods supported by System as specified in 3GPP AlarmIRP: CORBA Solution Set version 1:1.

*/

interface AlarmIRPOperations {

/*

This method returns Alarm Informations. If flag is TRUE, all returned Alarm Informations shall be in AlarmInformationSeq that contains 0,1 or more Alarm Informations. Output parameter iter shall be useless. If flag is FALSE, no Alarm Informations shall be in AlarmInformationSeq. IRPAgent needs to use iter to retrieve them.

*/

AlarmIRPConstDefs::AlarmInformationSeq get_alarm_list (

in string filter,

out boolean flag,

out AlarmInformationIterator iter

)

raises (GetAlarmList,ParameterNotSupported,InvalidParameter);

};

2.2.  Existing TS 32.106-6 Iterator

/**

Iterator is returned from find_managed_object.

The "id" field is for internal use, to keep track of the Iterator.

*/

valuetype Iterator {

private long id;

};

/**

The BasicCmIrpOperations interface.Supports a number of Resource Model versions.

*/

interface BasicCmIrpOperations {

/**

Performs a containment search, using a SearchControl to control the search and the returned results.

All MOs in the scope constitutes a set that the filter works on. The result Iterator contains all matched MOs, with the amount of detail specified in the SearchControl.

@parm baseObject The start MO in the containment tree.

@parm scope the SearchControl to use.

@raises UndefinedMOException The MO does not exist.

@raises IllegalDNFormatException The dn syntax string is malformed.

@raises IllegalScopeTypeException The ScopeType in scope contains an illegal value.

@raises IllegalScopeLevelException The scope level is negative (<0).

@raises IllegalFilterFormatException The filter string is malformed.

@raises IllegalCountException "howMany" has a value less than 0.

@see SearchControl

@see Iterator

*/

Iterator find_managed_objects(in DN baseObject,

in SearchControl scope

)

raises (UndefinedMOException,

IllegalDNFormatException,

UndefinedScopeException,

IllegalScopeTypeException,

IllegalScopeLevelException,

IllegalFilterFormatException,

IllegalCountException);

/**

Gets data from an Iterator.

@parm iter The Iterator to use. The "fetchedElements" field will be updated with the "howMany" (if possible) new values if "howMany" >= 0, where 0 means all remaining elements.

@parm howMany how many elements to return in the "fetchedElements" out parameter.

@parm fetchedElements the elements.

@returns A boolean indicating if any elements are returned."fetchedElements" is set to null when the Iterator is empty.

@raises UndefinedIteratorException The Iterator does not exist or is null.

@raises IllegalCountException "howMany" has a value < 0.

@see Iterator

*/

boolean get_next_elements(in Iterator iter,

in unsigned long howMany,

out Results fetchedElements)

raises (UndefinedIteratorException,

IllegalCountException);

/**

Deletes an Iterator.

@parm iter The Iterator to delete.

@raises UndefinedIteratorException The Iterator does not exist or is null.

@see Iterator

*/

void delete_iterator(in Iterator iter)

raises (UndefinedIteratorException);

};

2.3.  Proposal

It is proposed that the TS 32.106-6 iterator be changed to the iterator style used in TS 32.111-3.

Other proposed changes to TS 32.111-3 style iterator:

1. Clarify the IRPAgent’s responsibility for destroying the iterator

2. Clarify the IRPManager’s responsibility for destroying the iterator

3. Change how_many type from unsigned long to unsigned short. CORBA cannot support more than an unsigned short worth of alarms.

4. Clarify that less than how_many items may be returned, even if there are more items.

The proposed TS 32.111-3 are shown in the next section and document in a CR [1].

Proposed changes to TS 32.106-6 iterator and software:

1. Change the iterator to match TS 32.111-3 format

2. Add new iterator exceptions

3. Change find_managed_objects method to match TS 32.111-3 style methods using iterator

4. Delete get_next_elements and delete_iterator methods to instead use iterator

5. Remove UndefinedIteratorException exception

6. Remove InvalidCountException exception from find_managed_objects

7. Update comments accordingly

8. Update TS 32.106-6 accordingly

2.4.  Proposed New TS 32.111-3 Iterator

/**

The AlarmInformationIterator is used to iterate through a snapshot of Alarm Informations taken from the Alarm List when IRPManager invokes get_alarm_list. IRPManager uses it to pace the return of Alarm Informations.

IRPAgent controls the life-cycle of the iterator. However, a destroy operation is provided to handle the case where IRPManager wants to stop the iteration procedure before reaching the last iteration.

*/

interface AlarmInformationIterator {

        /**

This method returns between 1 and “how_many” Alarm Informations. The IRPAgent may return less than “how_many” items even if there are more items to send. “how_many” must be non-zero. Return TRUE if there are more Alarm Information to return. Return FALSE if there are no more Alarm Information to be returned. Note that the IRPAgent may both provide the last items in the alarm list and also indicate FALSE for completion.

If FALSE is returned, the IRPAgent will automatically destroy the iterator.

        */

        boolean next_alarmInformations (

          in unsigned short how_many,

          out AlarmIRPConstDefs::AlarmInformationSeq alarm_informations

        )

        raises (NextAlarmInformations,InvalidParameter);

        /**

        This method destroys the iterator.

        */

        void destroy ();

}; // end of AlarmInformationIterator

2.5.  Proposed New TS 32.106-6 Iterator

exception NextBasicCMInformations {

string reason;

};

exception InvalidParameter {

string reason;

};

/**

The BasicCMInformationIterator is used to iterate through a set of results on Basic CM operations.  Basic CM methods contains it as output parameter.

IRPManager uses it to pace the return of Basic CM operation results.

*/

interface BasicCMInformationIterator {

/**

This method returns between 1 and "how_many" Basic CM operation results. The IRPAgent may return less than “how_many” items even if there are more items to send. “how_many” must be non-zero. Return TRUE if there are more Basic CM operation results to return. Return FALSE if there is no more Basic CM operation results to be returned. Note that the IRPAgent may both provide the last items in the Basic CM operation results and also indicate FALSE for completion.

If FALSE is returned, the IRPAgent will automatically destroy the iterator. Otherwise, it is the IRPManager’s responsibility to destroy the iterator.

*/

boolean next_BasicCMInformations (

in unsigned short how_many,

out Results basicCM_informations

)

raises (NextBasicCMInformations,InvalidParameter);

/**

This method destroys the iterator.

*/

void destroy ();

}; // end of BasicCMInformationIterator

/**

The BasicCmIrpOperations interface. Supports a number of Resource Model versions.

*/

interface BasicCmIrpOperations {

/**

<version method not changed due to these proposed changes>

*/

/**

Performs a containment search, using a SearchControl to control the search and the returned results.

All MOs in the scope constitutes a set that the filter works on. The result Iterator contains all matched MOs, with the amount of detail specified in the SearchControl.

@parm baseObject The start MO in the containment tree.

@parm scope the SearchControl to use.

@parm flag TRUE if all results returned, FALSE id the iterator must be used

@parm iter Iterator used if flag is FALSE

@raises UndefinedMOException The MO does not exist.

@raises IllegalDNFormatException The dn syntax string is malformed.

@raises IllegalScopeTypeException The ScopeType in scope contains an illegal value.

@raises IllegalScopeLevelException The scope level is negative (<0).

@raises IllegalFilterFormatException The filter string is malformed.

@see SearchControl

@see BasicCMInformationIterator

*/

/*

This method returns Basic CM operation results. If flag is TRUE, all returned Basic CM operation results shall be in Results that contains 0, 1 or more Basic CM operation results. If flag is TRUE, output parameter iter shall be useless. If flag is FALSE, no Basic CM operation results shall be in Results. If flag is FALSE, IRPManager needs to use iter to retrieve them.

*/

Results find_managed_objects(

in DN baseObject,

in SearchControl scope,

out boolean flag,

out BasicCMInformationIterator iter

)

raises (UndefinedMOException,

IllegalDNFormatException,

UndefinedScopeException,

IllegalScopeTypeException,

IllegalScopeLevelException,

IllegalFilterFormatException);

};

2.6.  Outline Of TS 32-106-6 Clause Updates

1. Clause 5.5 – Iterator is no longer a valuetype.

2. Clause 6.2 – get_next_elements and delete_iterator no longer used.

3. Clause 6.3 – Additional out boolean flag and out BasicCMInformationIterator iter parameters.

4. Clause 6.3 – Changes to exceptions

5. Clause 6.3 – Changes due to that Results or BasicCMInformationIterator may be returned.
6. Clause 7 – Add the following paragraph before Table 12: Attributes that are denoted as “optional” may optionally be excluded from the OMG Structured Event. As an example, if the optional sourceIndicator attribute is not used for a particular notification, then the IRPAgent may exclude sourceIndicator from the  filterable body fields for that particular notification. Individual notifications from the same IRPAgent may include or exclude the same optional attribute.

NOTICE
This document has been prepared by Lucent Technologies Inc. (“Lucent”) to assist 3GPP subcommittee SA5. It is proposed to the subcommittee as a basis for discussion and is not to be construed as a binding proposal on Lucent. Lucent specifically reserves the right to amend or modify the material contained herein and nothing herein shall be construed as conferring or offering licenses or rights with respect to any intellectual property of Lucent.

